# $Id$
# Maintainer: Eric Bélanger <eric@archlinux.org>

pkgname=xine-ui
pkgver=0.99.10
pkgrel=1
pkgdesc="A free video player for Unix"
arch=('x86_64')
license=('GPL')
url="http://www.xine-project.org"
depends=('xine-lib' 'curl' 'libxft' 'libxxf86vm' 'lirc' 'xdg-utils')
options=('!strip')
source=(http://downloads.sourceforge.net/xine/${pkgname}-${pkgver}.tar.xz)
sha1sums=('4d61653d9bd88ba446c7c0772d03fc68ddfac46b')

build() {
  cd ${pkgname}-${pkgver}
  ./configure --prefix=/usr --mandir=/usr/share/man \
    --with-x --enable-lirc --without-aalib --enable-debug
  make
}

package() {
  cd ${pkgname}-${pkgver}
  make DESTDIR="${pkgdir}" desktopdir=/usr/share/applications install
}
